

PP Laboratorium 3.1

Moduły

Nowy moduł dodajemy do projektu klikając w menu: **Plik > Nowy moduł**. Kod nowego modu wygląda następująco:

```
unit Unit1;  
  
{$mode objfpc}{$H+}  
  
interface  
  
uses  
    Classes, SysUtils;  
  
implementation  
  
end.
```

Ponieważ nie będziemy programować obiektowo i nie wszystko jest nam potrzebne do pracy, należy usunąć zbędne sekcje kodu pozostawiając tylko sekcje **interface** oraz **implementation**:

```
unit Unit1;  
interface  
  
    {Tu będziemy umieszczali interfejs modułu}  
  
implementation  
  
    {Tu będziemy umieszczali implementację modułu}  
  
end.
```

UWAGA: Jeśli będziemy korzystać z dodatkowych modułów, należy dodać sekcję **uses** na samym początku interfejsu. Sekcja ta powinna zawierać tylko używane moduły, z których korzystamy w świadomy sposób. Pomiedzy **interface** a **uses** nie mogą się znajdować inne sekcje ani kod.

UWAGA: Podobnie jak program, moduł należy zapisać nadając mu nazwę. Nazwa modułu powinna spełniać wymagania wszystkich identyfikatorów. Najlepiej zawsze zaczynać nazwę modułu od przedrostka (prefiksu) **unit_**, **modul_** lub podobnego.

W interfejsie będziemy umieszczać wszystko to, co ma być widoczne poza modułem. Przede wszystkim wszystkie stałe, definicje własnych typów oraz nagłówki publicznych funkcji i procedur.

W sekcji implementacji umieszczamy definicje (ciała) funkcji i procedur, można tu też schować np stałe dostępne tylko w ramach modułu.

Przykładowy moduł zawierający definicję typu Matrix oraz procedury ReadMatrix i WriteMatrix:

```
unit unit_Matrix;
interface {##### INTERFACE #####}
const
    MATRIX_SIZE = 3;
type
    Matrix = Array [1..MATRIX_SIZE, 1..MATRIX_SIZE] of Real;

{Procedura wypisująca przekazaną macierz na ekranie}
procedure WriteMatrix(CONST macierz: Matrix);

{Procedura pobierająca macierz od użytkownika}
procedure ReadMatrix(VAR macierz: Matrix);

implementation {##### IMPLEMENTATION #####}

{Procedura wypisująca przekazaną macierz na ekranie}
procedure WriteMatrix(CONST macierz: Matrix);
var i,j: Integer;
begin
    for i:=1 to MATRIX_SIZE do
    begin
        for j:=1 to MATRIX_SIZE do
        begin
            Write(macierz[i,j]:6:1);
        end;
        WriteLn();
    end;
end;

{Procedura pobierająca macierz od użytkownika}
procedure ReadMatrix(VAR macierz: Matrix);
var i,j: Integer;
begin
    for i:=1 to MATRIX_SIZE do
    begin
        for j:=1 to MATRIX_SIZE do
        begin
            Write(i, ', ', j, ': ');
            ReadLn(macierz[i,j]);
        end;
    end;
end;

end. {#####}
```

Program laboratorium

- Moduły
 - Dodawanie własnego modułu
 - Sekcje w modułach
- Procedury i funkcje
 - Deklaracja procedury
 - Deklaracja funkcji
 - Przekazywanie parametrów (var, const)
 - Przekazywanie tablic (var, const)
 - Zwracanie wyniku funkcji